```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

We consider the differential time-delay model of a stirred tank studied in H. Kwakernaak, R. Sivan, *Linear Optimal Control Systems*, Wiley-Interscience, 1972. The system matrix is defined by:

```
> A:=DefineOreAlgebra(diff=[d,t],dual_shift=[delta,s],polynom=[t,s],
> comm=[theta,c[0],c[1],c[2],V[0]]):

> R:=matrix(2,4,[d+1/2/theta,0,-1,-1,0,d+1/theta,-(c[1]-c[0])/V[0]*delta,
> -(c[2]-c[0])/V[0]*delta]);
```

$$R := \begin{bmatrix} d+\frac{1}{2\,\theta} & 0 & -1 & -1 \\ 0 & d+\frac{1}{\theta} & -\frac{(c_1-c_0)\,\delta}{V_0} & -\frac{(c_2-c_0)\,\delta}{V_0} \end{bmatrix}$$

Let us consider the $A=\mathbb{Q}(c_0,c_1,c_2,V_0,\theta)[d,\delta]$-module $M=A^{1\times 4}/(A^{1\times 2}\,R)$ finitely presented by R. We compute the A-module structure of the endomorphism ring $E=\mathrm{end}_A(M)$ of M:

```
> Endo:=MorphismsConstCoeff(R,R,A):
```

We obtain that the A-module E is defined by

```
> nops(Endo[1]);
```

$$8$$

generators which satisfy

```
> rowdim(Endo[2]);
```

$$4$$

A-linear relations. We do not print the large outputs of *Endo*.

Let us now search for idempotents of E defined by two idempotent matrices $P\in A^{4\times 4}$ and $Q\in A^{2\times 2}$, i.e., P and Q satisfy the relations $R\,P=Q\,R$, $P^2=P$ and $Q^2=Q$:

```
> Idem:=IdempotentsMatConstCoeff(R,Endo[1],A,0,alpha):

> nops(Idem[1]);
```

$$12$$

We obtain 12 different matrices P satisfying the previous relations. Let us consider the first one where we have set to zero the two arbitrary constants:

```
> P:=subs(c81=0,c21=0,evalm(Idem[1,1])); Q:=Factorize(Mult(R,P,A),R,A);
```

$$P := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \frac{c_1-c_0}{c_1-c_2} & \frac{c_2-c_0}{c_1-c_2} \\ 0 & 0 & -\frac{c_1-c_0}{c_1-c_2} & -\frac{c_2-c_0}{c_1-c_2} \end{bmatrix} \quad Q := \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix}$$

As we have $P^2=P$ and $Q^2=Q$, we know that the A-modules $\ker_A(.P)$, $\mathrm{im}_A(.P)=\ker_A(.(I_4-P))$, $\ker_A(.Q)$ and $\mathrm{im}_A(.Q)=\ker_A(.(I_2-Q))$ are projective, and thus, free by the Quillen-Suslin theorem. Let us compute bases of those free modules:

```
> U1:=SyzygyModule(P,A): U2:=SyzygyModule(evalm(1-P),A):
> U:=stackmatrix(U1,U2);
```

$$U := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & c_1 - c_0 & c_2 - c_0 \end{bmatrix}$$

```
> X1:=SyzygyModule(Q,A): X2:=SyzygyModule(evalm(1-Q),A):
> X:=stackmatrix(X1,X2);
```

$$X := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

We then know that the matrix $U \in \mathrm{GL}_4(A)$ is such that the matrix R is equivalent to the following block-diagonal matrix $S = R\,U^{-1}$:

```
> S:=Mult(R,LeftInverse(U,A),A);
```

$$S := \begin{bmatrix} d + \frac{1}{2\,\theta} & -1 & 0 & 0 \\ 0 & 0 & d + \frac{1}{\theta} & -\frac{\delta}{V_0} \end{bmatrix}$$

We note that the second entry of the matrix S is invertible over A. Hence, we can use an elementary row operation to reduce the first row. In order to do that, we introduce the following unimodular matrix:

```
> X:=evalm([[0,1,0,0],[-1,d+1/(2*theta),0,0],[0,0,1,0],[0,0,0,1]]);
```

$$X := \begin{bmatrix} 0 & 1 & 0 & 0 \\ -1 & d + \frac{1}{2\,\theta} & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Then, the matrix $S\,X$ has the form:

```
> Mult(S,X,A);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & d + \frac{1}{\theta} & -\frac{\delta}{V_0} \end{bmatrix}$$

Hence, if we denote by $Y = X^{-1}\,U \in \mathrm{GL}_4(A)$ defined by

```
> Y:=Mult(LeftInverse(X,A),U,A);
```

$$Y := \begin{bmatrix} d + \frac{1}{2\,\theta} & 0 & -1 & -1 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & c_1 - c_0 & c_2 - c_0 \end{bmatrix}$$

then, the matrix R is equivalent to the following simple block-diagonal matrix $V\,R\,Y^{-1}$:

```
> Mult(V,R,LeftInverse(Y,A),A);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & d + \frac{1}{\theta} & -\frac{\delta}{V_0} \end{bmatrix}$$

Therefore, we have $M \cong A^{1\times 2}/(A\,(1 \quad 0)) \oplus A^{1\times 2}/(A\,(d + 1/\theta \quad \delta/V_0))$, i.e.:

$$M \cong A \oplus A^{1\times 2}/(A\,(d + 1/\theta \quad -\delta/V_0)).$$

If $\mathcal{F}$ denotes an A-module (e.g., $\mathcal{F} = C^\infty(\mathbb{R})$), then we obtain that the linear differential time-delay system $\ker_{\mathcal{F}}(R.)$ is equivalent to the linear system $\ker_{\mathcal{F}}(S.)$, i.e.:

$$\zeta_1 = 0, \quad \zeta_2 \in \mathcal{F}, \quad \dot{\zeta}_3(t) - \zeta_3(t)/\theta + \zeta_4(t-h)/V_0 = 0.$$

Let us study the A-module structure of the endomorphism ring $E = \mathrm{end}_A(M)$ of M:

```
> ext1:=Exti(Involution(Endo[2],A),A,1): ext1[1];
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

As we have $\mathrm{ext}^1_A(N, A) = 0$, where $N = A^{1\times 2}/(A^{1\times 4}\, R^T)$, because the previous matrix is the identity matrix (see F. Chyzak, A. Quadrat, D. Robertz, "OreModules: A symbolic package for the study of multidimensional linear systems", in the book *Applications of Time-Delay Systems*, J. Chiasson and J. -J. Loiseau (Eds.), Lecture Notes in Control and Information Sciences (LNCIS) 352, Springer, 233-264), we obtain that the A-module E is torsion-free. Let us check whether or not the A-module E is reflexive:

```
> ext2:=Exti(Involution(Endo[2],A),A,2)[1];
```

$$\begin{bmatrix} \delta & 0 & 0 & 0 \\ d\,\theta+1 & 0 & 0 & 0 \\ 0 & \delta & 0 & 0 \\ 0 & d\,\theta+1 & 0 & 0 \\ 0 & 0 & \delta & 0 \\ 0 & 0 & d\,\theta+1 & 0 \\ 0 & 0 & 0 & \delta \\ 0 & 0 & 0 & d\,\theta+1 \end{bmatrix}$$

As the previous matrix is not the identity matrix, we obtain that $\mathrm{ext}^2_A(N, A) \neq 0$, a fact proving that the A-module E is not reflexive. Hence, the A-module E is torsion-free but not free.

We proved that $M \cong A \oplus N$, where $N = A^{1\times 2}/(A\,(d + 1/\theta \quad \delta/V_0))$. Hence, we get:

$$E = \mathrm{end}_A(M) \cong \mathrm{end}_A(A) \oplus \mathrm{hom}_A(N, A) \oplus \mathrm{hom}_A(A, N) \oplus \mathrm{end}_A(N).$$

We have $\mathrm{end}_A(A) \cong A$ and $\mathrm{hom}_A(A, N) \cong N$. Let us compute the A-modules $\mathrm{hom}_A(N, A)$ and $\mathrm{end}_A(N)$. In order to do that, we introduce the two matrices $Z = 0$ and $T = (d + 1/\theta \quad \delta/V_0)$:

```
> Z:=evalm([[0]]);
```

$$Z := \begin{bmatrix} 0 \end{bmatrix}$$

```
> T:=submatrix(S,2..2,3..4);
```

$$T := \begin{bmatrix} d + \frac{1}{\theta} & -\frac{\delta}{V_0} \end{bmatrix}$$

Let us check that we have $\mathrm{hom}_A(A, N) \cong N$ by computing the A-module $\mathrm{hom}_A(A, N)$:

```
> E[1]:=MorphismsConstCoeff(Z,T,A);
```

$$E_1 := [[\begin{bmatrix} 1 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 1 \end{bmatrix}], \begin{bmatrix} V_0\, d\,\theta + V_0 & -\delta\,\theta \end{bmatrix}]$$

We obtain that the A-module $\mathrm{hom}_A(A, N)$ is defined by two generators f_1 and f_2 satisfying the A-linear relation $V_0\,\theta\, d\, f_1 - \theta\,\delta\, f_2 = 0$, which is precisely the definition of the A-module N.

Let us now compute the A-module $\hom_A(N, A)$:

```
> E[2]:=MorphismsConstCoeff(T,Z,A);
```

$$E_2 := [[\left[\begin{array}{c} \delta\,\theta \\ V_0\,d\,\theta + V_0 \end{array}\right]], []]$$

We obtain that A-module $\hom_A(N, A)$ is defined by one generator which does not satisfy any A-linear relation, i.e., it is not a torsion element. Hence, we obtain $\hom_A(N, A) \cong A$. We note that we have $T\,E[2][1] = 0$, which is consistent with the fact that $\hom_A(N, A) \cong \ker_A(T.)$.

Let us compute the A-module $\mathrm{end}_A(N)$:

```
> E[3]:=MorphismsConstCoeff(T,T,A):
> E[3][1];
```

$$[\left[\begin{array}{cc} 0 & \delta\,\theta \\ 0 & V_0\,d\,\theta + V_0 \end{array}\right], \left[\begin{array}{cc} 1 & 0 \\ 0 & 1 \end{array}\right]]$$

The A-module $\mathrm{end}_A(N)$ is defined by two generators g_1 and $g_2 = \mathrm{id}_N$ which satisfy the relation:

```
> E[3][2];
```

$$\left[\begin{array}{cc} -1 & V_0\,d\,\theta + V_0 \end{array}\right]$$

Hence, we obtain that $g_1 = V_0\,(\theta\,d + 1)\,\mathrm{id}_M$, i.e., the A-module $\mathrm{end}_A(N)$ is generated by id_M, which proves that $\mathrm{end}_A(N) \cong A$. We finally obtain:

$$E \cong A \oplus N \oplus A \oplus A = A^3 \oplus N.$$

We can check the previous result by studying the A-module E. One way to do that is to find an injective A-morphism $\varphi : N \longrightarrow E$ such that $\mathrm{coker}\,\varphi \cong A^{1\times 3}$. Using OreMorphisms, we can try to handle the corresponding computations. We first compute the A-module $\hom_A(N, E)$:

```
> Morph:=MorphismsConstCoeff(T,Endo[2],A):
```

If we consider the matrix $P = Morph[1][6] \in A^{2\times 8}$ defining the element h of $\hom_A(N, E)$ and compute a matrix $Q \in A^{1\times 8}$ satisfying $T\,P = Q\,Endo[2]$,

```
> Y:=Morph[1][6]; Z:=Factorize(Mult(T,Y,A),Endo[2],A);
```

$$Y := \left[\begin{array}{cccccccc} 0 & 0 & 0 & V_0\,\theta\,(-c_1 + c_2) & 0 & 0 & 0 & 0 \\ \theta\,V_0\,(c_2\,c_1 - c_2\,c_0 + {c_0}^2 - c_1\,c_0) & -\theta\,V_0\,(c_2\,c_1 - c_2\,c_0 + {c_0}^2 - c_1\,c_0) & 0 & 0 & \theta\,V_0\,(-2\,c_2\,c_0 + {c_0}^2 + {c_2}^2) & 0 & 0 & 0 & -\theta\,V_0\,(-2\,c_1\,c_0 + {c_0}^2 + {c_1}^2) \end{array}\right]$$

$$Z := \left[\begin{array}{ccccc} 0 & 0 & 0 & 0 & 1 \end{array}\right]$$

then we can compute the A-module $\ker\varphi$:

```
> K:=KerMorphism(T,Endo[2],Y,Z,A);
```

$$K := [[\,1\,], [\,V_0\,d\,\theta + V_0 \quad -\delta\,\theta\,], [\,d + \theta^{-1} \quad -\tfrac{\delta}{V_0}\,]], [[\,\tfrac{1}{\theta\,V_0}\,]]$$

As the first matrix $K[1][1]$ is 1, we obtain that $\varphi \in \hom_A(N, E)$ is injective.

Let us now compute the A-module $\mathrm{coker}\,\varphi$:

```
> Coker:=CokerMorphism(T,Endo[2],Y,Z,A):
```

Let us compute its rank:

```
> OreRank(Coker,A);
```

$$3$$

We finally need to check whether or not the A-module $\operatorname{coker}\varphi$ is free:

```
> Ext1:=Exti(Involution(Coker,A),A,1):  Ext1[1];
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

We obtain that the A-module $\operatorname{coker}\varphi$ is torsion-free. Moreover, we have $\operatorname{coker}\varphi \cong A^{1\times 8}\, Ext1[3]$.

```
> map(factor,LeftInverse(Ext1[3],A));
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & \frac{1}{2\,(c_2-c_1)^2 V_0\,\theta} & \frac{1}{(c_2-c_1)^2(c_1-c_0)} & \frac{1}{(c_2-c_1)^2(c_0-c_1)} & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{c_2-c_0}{2\,(-c_1+c_2)^2\,V_0\,\theta} & \frac{c_0-c_2}{(c_2-c_1)^2(c_0-c_1)} & -\frac{1}{(c_2-c_1)^2} & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{c_0-c_2}{2\,V_0\,\theta\,(c_2-c_1)} & \frac{1}{c_1-c_2} & \frac{c_0-c_2}{(c_1-c_2)(c_1-c_0)} & 0 \end{bmatrix}$$

As the matrix $Ext1[3]$ admits a left-inverse over A, we obtain that $\operatorname{coker}\varphi \cong A^{1\times 3}$, which proves that we have the split exact sequence of A-modules $0 \longrightarrow N \xrightarrow{\varphi} E \longrightarrow A^{1\times 3} \longrightarrow 0$, a fact implying that $E \cong N \oplus A^3$ and proves the result. In all the previous computations, we have assumed that we were in the generic situation, i.e., the constants c_0, c_1 and c_2 are pairwise different. As the module properties of M are known to depend on the system parameters (see F. Chyzak, A. Quadrat, D. Robertz, OreModules project, `http://wwwb.math.rwth-aachen.de/OreModules` for the precise details), we let the reader handle the different non-generic situations.